
; **********************************************************************
; Diesel Engine Meter- A program for measuring timing patterns on a
;                      diesel engine flywheel.  An optical interrupter
;                      switch is used to observe rotating ýns, while
;                      timing data is stored in a buffer.  When 320 
;                      datapoints have been taken, the data is sent
;                      over 2400 baud serial to a host computer.  Two
;                      LEDs are used for status monitoring.  The green
;                      LED blinks in sync with switch events, and the
;                      red LED toggles when the data buffer is full and
;                      output is generated.
;
; Chris Lyon, scooby@mit.edu
; Isaac Chuang, ichuang@media.mit.edu
; Amy Sun, asun@media.mit.edu
;
;     Needed:  
;      PIC Foundation
;            Proto Layer
;            RS-232 Module
;     Attach:
;            Pin B2 - Green LED
;            Pin B3 - Red LED
;            Pin C2 - Optical Interrupter (this pin needed for capture)
; **********************************************************************

; Global Constant Declarations 
; basic conýg 
    [const @ 0] 
    [const timer 1] 
    [const pcl 2] 
    [const status 3] 
    [const c 0][const z 2][const bankl 5][const bankh 6][const ibank 7] 
    [const @@ 4] 
    [const porta 5][const porta-ddr $85] 
    [const portb 6][const portb-ddr $86] 
    [const portc 7][const portc-ddr $87] 
    [const portd 8][const portd-ddr $88] 
    [const porte 9][const porte-ddr $89] 
    [const option 1] ; bank 1 
; ad conversion 
    [const adres $1e] 
    [const adcon0 $1f] 
    [const adcon1 $1f] ;bank 1 

    [const adon 0][const adgo 2] 
; interrupts 
    [const intcon $0b] 
    [const gie 7][const peie 6][const t0ie 5][const inte 4] 
    [const intf 1][const t0if 2] 
    [const pir1 $0c] 
    [const rcif 5][const txif 4][const sspif 3][const ccp1if 2] 
    [const pie1 $0c] ; bank 1 
    [const ccp1ie 2] 
; timer 
    [const tmr1l $0e] 
    [const tmr1h $0f] 
; flash 
    [const eedatah $0e][const eedatal $0c] ;bank 2 
    [const eeaddrh $0f][const eeaddrl $0d] ;bank 2 
    [const eecon1 $0c] ; bank 3 
    [const rd 0][const wr 1][const wren 2][const eepgd 7] 
    [const eecon2 $0d] ;bank 3 ; timer 
    [const t1con $10] 
; ssp 
    [const sspbuf $13] 
    [const sspcon1 $14] ; bank 0 
    [const sspen 5][const sspm3 3] 
    [const sspcon2 $11] ; bank 1 
    [const gcen 7][const ackstat 6][const ackdt 5][const acken 4] 
    [const rcen 3][const pen 2][const rsen 1][const sen 0] 
    [const sspadd $13] ; bank 1 
    [const sspstat $14] ; bank 1 
    [const smp 7][const bf 0] 
; capture 
    [const ccpr1l $15] 
    [const ccpr1h $16] 
    [const ccp1con $17] 
; uart 
    [const baud $19] ; bank 1 
    [const txctrl $18] ; bank 1 
    [const rxctrl $18] 
    [const txreg $19] 
    [const rxreg $1a] 

; ram locations - all in the shared high bytes of the ram banks 
; these are our primary local variables 
    [const tmp $78] ; tmp 16 bit register - low byte 
    [const tmp2 $79] ; tmp 16 bit register - high byte 
    [const mybank $7a] ; what bank weôre using for the data buf 
    [const data $7b] ; data - low byte 
    [const data2 $7c] ; data - high byte 
    [const counter $7d] ; general purpose counter - used in putdec 

    [const int-a $7e] 
    [const int-status $7f] 

;----------------------------------------------------------------------- 
; performance parameters 
    [const capture-mode $05] ; trig on every rising edge 
    [const timer-mode $31] ; div by 8 = 4 us ((8mhz/4)/8)) 

;----------------------------------------------------------------------- 
; Main program: starts here 
start 
    [bsr io-init] ; run io-init 
    [bsr greeting] 
mainloop ; inýnite loop - we skip the middle man! 
    [bra mainloop] ; program is entirely interrupt driven 

;----------------------------------------------------------------------- 
; subroutine: print greeting, with useful parameters 
greeting 
    [ldan $65] ; óeô 
    [bsr serial-send] 
    [ldan $33] ; ó3ô 
    [bsr serial-send] 
    [ldan $3a] ; ó:ô 
    [bsr serial-send] 
    [ldan $20] ; ó ó 
    [bsr serial-send] 
    [ldan capture-mode] 
    [bsr puthex] 
    [ldan timer-mode] 
    [bsr puthex] 
    [bra crlf] 

;----------------------------------------------------------------------- 
; subroutine: test routines 
test-mem ; should store 0-255 then dump 
    [ldan $20] ; initial address for data buffer 
    [sta @@] ; store in FSR 
    [ldan $0] ; initial bank 
    [sta mybank] 
    [ldan 0] 
    [sta data] 
tm_loop 
    [lda data] ; ýll data buf with 0-255 

    [bsr push-byte] 
    [incsz data] 
    [bra tm_loop] 
    [bra dumpdat] 
test-pd16 
    [ldan $30] ; should output 12345 
    [sta data2] 
    [ldan $39] 
    [sta data] ; $3039 = 12345 
    [bsr putdec16] 
    [ldan $00] ; 
    [sta data2] 
    [ldan $00] 
    [sta data] ; should output zero 
    [bra putdec16] 
test-putdec ; output 000-255 in decimal 
    [ldan 0] ; test routine for putdec 
    [sta mybank] 
tpd_lp [lda mybank] ; output in hex 
    [bsr putdec] ; 
    [ldan 32] ; output space 
    [bsr serial-send] 
    [incsz mybank] ; increment, just once through 
    [bra tpd_lp] 
    [bra mainloop] 

;----------------------------------------------------------------------- 
; serial output routine 
serial-send 
    [btss txif pir1] 
    [bra serial-send] ; wait until the serial out buffer is clear 
    [sta txreg] ; store the value in the tx holding register 
    [rts] 

;----------------------------------------------------------------------- 
; I/O initialization 
io-init 
    [bclr 2 portb] ; clear both led pins 
    [bclr 3 portb] 
    ; buffer pointer init 
    [ldan $20] ; initial address for data buffer 
    [sta @@] ; store in FSR 
    [ldan $0] ; initial bank 
    [sta mybank] 
    [bset bankl status] ; set led pins as outputs 
    [bclr 2 portb] 

    [bclr 3 portb] 
    ; serial port init 
    [ldan 51][sta baud] ; 51 = 2400 baud (@8mhz) ; 12 = 9600 
    [ldan $20][sta txctrl] ; enable the transmitter 
    [bclr bankl status] 
    [ldan $90][sta rxctrl] ; and the receiver 
    ; timer init 
    [clr tmr1l] ; reset timer 1 module 
    [clr tmr1h] 
    [ldan timer-mode] ; set timer 1 prescaler & turn on 
    [sta t1con] ; should give resolution of 4 us ((8mhz/4)/8)) 
    ; capture module init 
    [ldan capture-mode] ; enable compare mod1 
    [sta ccp1con] 
    [bset bankl status] ; bank 1 
    [bset ccp1ie pie1] ; enable capture interrupt 
    [bclr bankl status] ; bank 0 
    [bclr ccp1if pir1] ; clear the capture interrupt flag 
    [bset peie intcon] ; enable peripheral interrupts (Capture is one) 
    [bset gie intcon] ; general interrupt enable 
    [rts] 

;----------------------------------------------------------------------- 
; interrupt routine 
int-routine ; If we made it here, weôve triggered a rising edge 
    [sta int-a] ; save the accumulator 
    [lda status] 
    [sta int-status] ; save status 
    [bclr bankl status] ; make sure weôre in bank 0 
    [bclr bankh status] 
    [lda portb] ; toggle green LED every pulse 
    [xorn $04] 
    [sta portb] 
    [lda ccpr1h] ; get high byte of timer1, and save in data buf 
    [bsr push-byte] 
    [lda ccpr1l] 
    [bsr push-byte] ; get low byte of timer1, and save in data buf 
    ; is the data buffer full? check mybank for overflow 
    [btss 2 mybank] ; if bit 2 = 1 then overflow: end of buffer 
    [bra iret] ; not overflow - go to interrupt return 
    ; we have a full buffer - dump to serial 
    [lda portb] ; toggle red LED to indicate buffer full 
    [xorn $08] 
    [sta portb] 
    [ldan $20] ; start address for data buffer 
    [sta @@] ; store in FSR 
    [ldan $0] ; initial bank 

    [sta mybank] 
    [bsr dumpdat] ; dump data to serial port 
iret ; interrupt return 
    [clr tmr1l] ; clear timer1 registers 
    [clr tmr1h] 
    [bclr ccp1if pir1] ; clear the interrupt flag 
    [lda int-status] ; restore status and accumulator registers 
    [sta status] 
    [swap int-a] ; tricky way of doing an lda without 
    [lswap int-a] ; affecting the status 
    [rti] 

;----------------------------------------------------------------------- 
; subroutine: dump data buffer to serial port 
dumpdat 
    [ldan $20] ; start address for data buffer 
    [sta @@] ; store in FSR 
    [ldan $0] ; initial bank 
    [sta mybank] 
ddloop 
    [bsr read-byte-inc] ; read byte (high) 
    [sta data2] ; save as high byte 
    [bsr read-byte-inc] ; read byte (low) 
    [bsr putdec16] ; output to serial port 
    [ldan 32] ; send space 
    [bsr serial-send] 
    [btss 2 mybank] ; if bit 2 = 1 then end of buffer 
    [bra ddloop] ; not overflow - keep looping 
    [ldan $20] ; initial address for data buffer 
    [sta @@] ; store in FSR 
    [ldan $0] ; reset the buffer pointer 
    [sta mybank] 
crlf 
    [ldan 10] ; send crlf 
    [bsr serial-send] 
    [ldan 13] 
    [bra serial-send] 

;----------------------------------------------------------------------- 
; subroutine: push byte onto data buffer and increment pointer 
push-byte 
    [bclr ibank status] ; set bank according to lowest two 
    [bclr 7 @@] ; bits of mybank (status.ibank and fsr.7) 
    [btsc 0 mybank] 
    [bset 7 @@] ; fsr.7 is low bit of bank 
    [btsc 1 mybank] 

    [bset ibank status] ; status.7 (IRP) is high bit of bank 
    [sta @] ; store data in location pointed to by FSR 
inc-ptr 
    [bclr 7 @@] ; clear bit 7 so we can detect end of this bank 
    [inc @@] ; increment FSR 
    ; $20 = 0010 0000, $70 = 0111000 
    [btss 6 @@] ; test for $70 overflow (only use $20-$6f) 
    [rts] ; return to caller now if FSR.6 is clear 
    [btss 5 @@] ; return to caller now if FSR.5 is clear 
    [rts] ; [btss 4 @@] ; return to caller now if FSR.4 is clear 
    [rts] ; 
    [ldan $20] ; reset FSR 
    [sta @@] 
    [inc mybank] ; move to next bank (only bits 0,1 matter) 
    [rts] ; return to caller 

;----------------------------------------------------------------------- 
; subroutine: read byte from buffer and inc pointer - return in data 
read-byte-inc 
    [bclr ibank status] ; set bank according to lowest two 
    [bclr 7 @@] ; bits of mybank (status.ibank and fsr.7) 
    [btsc 0 mybank] 
    [bset 7 @@] ; fsr.7 is low bit of bank 
    [btsc 1 mybank] 
    [bset ibank status] ; status.7 (IRP) is high bit of bank 
    [lda @] ; store data in location pointed to by FSR 
    [sta data] 
    [bra inc-ptr] ; branch to increment pointer 

;----------------------------------------------------------------------- 
; subroutine: hex output function 
puthex 
    [sta data] ; save the value 
    [sta data2] ; save a second copy 
    [ror data2] ; rotate through 4 times to right to grab high nibble 
    [ror data2] ; should be able to use swap here 
    [ror data2] 
    [lror data2] 
    [bsr nibble-out] ; send the high nibble 
    [lda data] ; reload the value 
    ; send the low nibble 
nibble-out 
    [andn $0f] ; strip off high nibble 
    [sta data2] ; save it 
    [subn 9] ; are we greater than 9? 
    [lda data2] ; reload the accumulator with the data 

    [btss c status] ; if weôre greater than 9 
    [addn 7] ; add 7 to the ascii value will be a/b/c/d/e/f 
output-num 
    [addn 48] ; add 48 to get the ascii decimal value 
    [bsr serial-send] ; send out the byte 
    [rts] 

;----------------------------------------------------------------------- 
; subroutine: decimal output function - call with data in accumulator 
; algorithm: 
; cnt = 0; 
; while(x >= 100) { cnt++; x -= 100; } 
; print cnt; 
; cnt = 0; 
; while(x >= 10) { cnt++; x -= 10; } 
; print cnt; 
; print x; 
putdec 
    [sta data] ; save data 
    [ldan 0] ; zero counter 
    [sta counter] ; 
    [ldan 100] ; start with hundredôs digit 
    [bsr pd_digit] 
    [ldan 10] ; tenôs digit 
    [bsr pd_digit] 
    [lda data] 
    [bra pd_dig2] ; output ýnal oneôs digit 
pd_digit ; output digit, number to sub in acc 
    [subm data] ; subtract acc from data and store in data 
    [btsc c status] ; C=0 means result negative 
    [bra pd_pos] ; result was positive 
    [addm data] ; result was negative: add back 
    [lda counter] ; output counter value 
pd_dig2 
    [addn $30] 
    [bsr serial-send] ; send to output 
    [ldan 0] ; zero counter 
    [sta counter] ; 
    [rts] 
pd_pos 
    [inc counter] ; increment counter 
    [bra pd_digit] ; loop 

;----------------------------------------------------------------------- 
; 16-bit subtraction: subtract tmp from data 
sub16 

    [lda tmp] 
    [subm data] ; data -= tmp 
    [btss c status] ; C=0 means result negative 
    [bra sub16c] ; negative! handle carry 
    [lda tmp2] ; positive! no carry to high byte 
    [subm data2] ; data2 -= tmp2 [rts] 
sub16c 
    [inc tmp2] ; works for our limited cases of interest 
    [lda tmp2] ; and the resulting carry is right 
    [subm data2] ; data2 -= tmp2+1 
    [dec tmp2] 
    [rts] 

;----------------------------------------------------------------------- 
; 16-bit addition: add tmp to data 
add16 
    [lda tmp] 
    [addm data] ; data += tmp 
    [btsc c status] ; C=1 means carry needed 
    [inc data2] ; carry set! handle carry by inc 
    [lda tmp2] ; no carry to high byte 
    [addm data2] ; data2 += tmp2 
    [rts] 

;----------------------------------------------------------------------- 
; print out 16-bit decimal number - call with the value in [data2,data] 
putdec16 
    [ldan 0] ; zero counter 
    [sta counter] ; 
    [ldan $27] ; start with 10000 digit 
    [sta tmp2] 
    [ldan $10] 
    [sta tmp] ; $2710 = 10,000 
    [bsr pd16_digit] 
    [ldan $3] ; next: 1000 digit 
    [sta tmp2] 
    [ldan $e8] 
    [sta tmp] ; $3e8 = 1,000 
    [bsr pd16_digit] 
    [ldan $0] ; next: 100 digit 
    [sta tmp2] 
    [ldan 100] 
    [sta tmp] 
    [bsr pd16_digit] 
    [ldan 10] ; tenôs digit 
    [bsr pd_digit] 

    [lda data] 
    [bra pd_dig2] ; output ýnal oneôs digit 
pd16_digit ; output digit, number to sub in [tmp2,tmp] 
    [bsr sub16] ; subtract tmp from data and store in data 
    [btsc c status] ; C=0 means result negative 
    [bra pd16_p] ; result was positive 
    [bsr add16] ; result was negative: add back 
    [lda counter] ; output counter value 
    [addn $30] 
    [bsr serial-send] ; send to output 
    [ldan 0] ; zero counter 
    [sta counter] ; 
    [rts] 
pd16_p 
    [inc counter] ; increment counter 
    [bra pd16_digit] ; loop 


